

```

//Zahlenbeispiele zum Artikel http://de.wikipedia.org/wiki/Zeitgleichung Abschnitt "Analytische Berechnung"
//Der Code ist als sce-Datei in Scilab ( http://www.scilab.org/scilab/about )ausführbar.
//Stand: 08.07.2013, 14:45 OEZ
//http://commons.wikimedia.org/wiki/File:ZeitgleichungAnalytisch.pdf
//[[File:ZeitgleichungAnalytisch.pdf|thumb|Analytische Berechnung der Zeitgleichung bei Vorgabe der exzentrischen Anomalie]]
clear, clc(), mode(0),format(14) //"Präambel"

function [ZG, V, t_, M, otr, VW]=!ZG(E, e, eps, omega, Jtr) //Zeitgleichung ZG in Funktion der exzentrischen Anomalie E
//Bezeichnungen und Eiheiten nach Artikelabschnitt
//ZG, V, t_, M sind Zeilenvektoren von gleicher Länge wie E.
VW=%pi/2-omega //wahre Anomalie bei Winteranfang (negativer Wert)
otr=2*%pi/Jtr //mittlere Winkelgeschwindigkeit der Erd-Translation
t_=(E-e*sin(E))/otr //Zeit ab Periheldurchgang
M=otr*t_ //mittlere Anomalie
V=!atan3(sqrt(1-e^2)*sin(E)/(cos(E)-e),ones(E),E)//wahre Anomalie
ZG=1/2*%pi*(otr*t_-VW-!atan3(tan(V-VW)/cos(eps),ones(E),V-VW))//Zeitgleichung (in Tagen)
endfunction

function w=!atan3(y, x, wNah)//Arcustangens-Wert nächst wNah
//y: Ordinate, Typ Matrix
//x: Abszisse, Typ wie y
//wNah: Winkel,Skalar oder Typ wie y
//w: Winkel, Typ wie y
w=atan(y,x)
w=w+%pi*round( (wNah-w)/%pi )
endfunction

GRD=%pi/180; //Urechnung von rad nach °
e=0.0167//numerische Exzentrizität der Erdbahn
eps=23.44*GRD //Schiefe der Erdachse
omega=103*GRD //Argument des Perihels (Winkel der Linie Sonne->Erde im Perihel gegen die Linie Sonne->Erde am Winteranfang)
Jtr=365.24 //Umlaufperiode (Translationsperiode, tropisches Jahr)
E=linspace(0,2*%pi,100); //exzentrische Anomalie während eines Jahres, 3. Argument ist Anzahl der Werte
//#####
[ZG,V,t_,M,otr,VW]=!ZG(E,e,eps,omega,Jtr); //Funktionsaufruf
//#####
xdel(); //löscht vorhandene Graphik
plot(t_,ZG*60*24) //Zeitgleichung
xtitle("Zeitgleichung","t_/Tage","ZG/Minuten")
cax=gca(); cax.grid=[1 1]; //Gitternetz

//Fortsetzung auf S. 2

```

```

//Zeitgleichung bei Perihel und Aphel:
ZG_AP=1/2/%pi*(-VW+!atan3(tan(VW)/cos(eps),1,-VW))*24*60; printf('\n Zeitgleichung für Aphel und Perihel: %5.2f min',ZG_AP);
//Ergebnis: [-4.50 min -4.50 min]

//Zeitgleichung am 30. Jan. 2012, 12:00 MEZ, Index B(eispiel)
printf("\n")
//Als Zeitbezug wird der Winteranfang 2011 gewählt, Perihel ungeeignet wegen starken Mondeinflusses
EW=2*!atan3(sqrt((1-e)/(1+e))*tan(VW/2),ones(VW),VW/2)//exzentrische Anomalie für Winteranfang im Keplermodell//-0.2231662
tW_=(EW-e*sin(EW))/otr//Winteranfangs-Zeit ab Perihel im Keplermodell//-12.757746
tW=datenum(2011,12,22,5,30,00)//Winteranfang 2011 nach MICA (UT) umgeformt in Zeit tW ab 1-Jan-0000//734859.229167
tP=tW-tW_//Periheldurchgang gemäß KEPLERmodell als Zeit ab 1-Jan-0000//734871.986913//. Bitte nicht mit wahrem Perihel
verwechseln!
[Y,M,D,H,MI,S]=datevec(tP)////Periheldurchgang gemäß KEPLERmodell//3.1.2012 23:41:9.3 UT. Bitte nicht mit wahrem Perihel
verwechseln!
tB=datenum(2012,1,30,11,00,0.0)//UT, Datum des Referenz-Zahlenbeispiels als UT-Zeit ab 1-Jan-0000 //734898.458333
t_B=tB-tP//Zeit ab Perihel//26.471420
ZG_B=interp(t_B ,t_,ZG, splin(t_,ZG))*24*60//Zeitgleichung bei t_B, interpoliert aus Werten des Aufrufs von !ZG(...) oben
// Ergebnis:
// Zeitgleichung ZG_B für 30. Jan. 2012, 12:30 MEZ (t_B): -13.21 min

//Zeitgleichung bei den Jahreszeiten-Anfängen:
printf("\n")
V=VW+[1:4]*%pi/2;
E=2*!atan3(sqrt((1-e)/(1+e))*tan(V/2),ones(V),V/2);
[ZG]=!ZG(E, e, eps, omega, Jtr);
ZG_=ZG*24*60; printf("\n ZG_F: %4.2f min, ZG_S: %4.2f min, ZG_H: %4.2f min, ZG_W: %4.2f min ",ZG_)
// Ergebnisse:
// ZG_F: -7.44 min, ZG_S: -1.74 min, ZG_H: 7.48 min, ZG_W: 1.70 min

```